DESIGNED FOR HP-IB SYSTEMS

Programming Note

8505A/9826A-1
Supersedes: None

OCTOBER 1981

Introductory Operating Guide
for the HP 8505A RF Network Analyzer with the HP 9826A Desktop Computer (BASIC)





INTRODUCTION

This programming note is a guide to the remote operation of the HP 8505A RF Network Analyzer, the HP 8501A Storage Normalizer, and the HP 8503A S-Parameter Test Set using the HP 9826A Desktop Computer and the BASIC programming language. The 8505A is a fully programmable 500 kHz-1.3 GHz vector network measurement test system with 100 dB dynamic range. Frequency, magnitude, phase, and group delay can be measured with resolutions up to 100 Hz, 0.01 dB, 0.1°, and 0.1nsec respectively. For better frequency resolution and stability, the internal source has an optional phase lock capability for measuring difficult devices like crystal filters and SAW devices. When the HP 8660C Synthesized Signal Generator (with the appropriate 86600 Series Plug-in) is used for phase lock operation, one hertz center frequency resolution can be achieved. The 8501A Storage Normalizer enhances measurements with digital storage, normalization, signal averaging, CRT labeling, and graphics capability.

This measurement system is fully compatible with the Hewlett-Packard Interface Bus (HP-IB). The HP 9826A Desktop Computer is an ideal instrument controller featuring two times the HP 9825A's average execution speed, in BASIC language, with mini-disc mass storage, a two-mode graphics display, built-in standard HP-IB interface, and expandable memory and interfacing. This note demonstrates the HP-IB control of the 8505A via the BASIC programming language.

Included in this guide are the system connections for remote operation, and a modular sample program that demonstrates the capabilities of the system and provides a starting point to those who want to create programs tailored to their specific measurement needs.

Reference Information

For further information on the HP Interface Bus, the following should prove helpful:

Tutorial Description of the Hewlett-Packard Interface Bus (HP Literature No. 5952-0156).

Complete reference information on the 8505A, the 8501A, and the 8503A can be found in their respective Operating and Service (O/S) manuals. The HP part number for each is as follows:

O/S Manual 8505A: 08505-90002
O/S Manual 8501A: 08501-90022

HEWLETT PACKARD

O/S Manual 8503A: 08503-90001
8505A Programming Code Summary: HP Literature No. 5952-9293.

For information on operating the 9826A the following references are available:

9826A Operating Manual (HP Part No. 09826-90000).
BASIC Language Reference (HP Part No. 09826-90055).
BASIC Programming Manual (HP Part No. 09826-90010).
BASIC I/O Manual (HP Part No. 09826-90020).

Equipment Required

To perform all the example programs described in this programming note, you will need the following equipment and accessories:

HP 8505A Network Analyzer with Option 007 (Labeling Interface)
HP 8501A Storage Normalizer
HP 8503A S-Parameter Test Set
HP 11857A Test Port Extension Cables
HP 85031A APC-7 Calibration Kit
HP 9826A Desktop Computer with Opt. 001 (or 701)—BASIC programming language
A 200 MHz Band-pass Filter or an equivalent test device

For hard copy output:

A Thermal Graphics Printer which is compatible with the HP 9826A (HP 2673A)

For the phase lock option:

HP 8660C Synthesized Signal Generator
HP 11667A Power Splitter
HP 8491A Opt. 003 3dB Attenuator

Set-Up

Figure 1 shows the system connection. All that is required is to:

1. Connect the 24-pin HP-IB connector of the built-in HP-IB interface of the 9826A to the printer and the system consisting of 8501A, 8505A, and 8503A. Refer to O/S Manuals for more detail on interconnects.

CAUTION

Do not attempt to mate black metric threaded screws on one connector with silver English threaded nuts on another connector, or vice versa, as damage may result. A metric conversion kit which will convert one cable and one or two instruments to metric hardware is available by ordering HP Part No. 5060-0138.

2. Following the instructions in the BASIC Operating Manual for the HP 9826A, load the BASIC language system.
3. The HP-IB select code is assumed to be preset to 7. This programming note assumes the following instrument HP-IB addresses:

Instrument	Address
8505A Source/Converter	19
8505A Processor/Display	16
8501A Storage-Normalizer	14
8503A Test Set	20
8660C Synth. Sig. Gen.	18
2673A Graphics Printer	01



Figure 1. System Connection

You may change the above numbers according to your instrument addresses. If there is a conflict, these addresses may be modified by removing instrument covers and manipulating appropriate slide switches or jumpers on various circuit cards. For detailed instructions, see Chapter E of the 8505A Operating and Service Manual.

Check-Out

If your computer has a soft-loaded operating system and power had been turned off, then follow instructions to reload the system. When the message "BASIC READY" reappears, turn on the rest of the equipment.

To verify that the HP-IB connections and interface are functional execute the following program:

```
10 ! Define instrument addresses
11 !
20 Normalizer=714
30 Processor=716
40 Source=719
50 Synthesizer=718
60 Testset=720
61 !
70 ! System "CHECK-OUT"
71 !
80 REMOTE Processor
90 DISP "REM: Sig. Proc-8505"
100 PAUSE
110 REMOTE Source
120 DISP "REM: Source-8505"
130 PAUSE
140 REMOTE Normalizer
150 DISP "REM: Stor. Norm.-8501"
160 PAUSE
170 REMOTE Testset
180 DISP "REM: Test Set-8503"
190 BEEP
200 PAUSE
```

Verify that the appropriate REMote light is lit according to the displayed messages before pressing CONTINUE to proceed. If the 9826A display indicates an error, check the program statement and the system connections and execute the program again.

Before programming the 8505A system for different functions, the user should be aware of the extent of remote control that can be used. The Remote Enable ('REMOTE') command sets the instruments into remote control from the local (manual) mode. In remote, the instruments will perform only as their functions are programmed. Any controls not programmed will assume their "front panel" state (as manually postioned).

If the LOCAL buttons are pressed, the instruments will return to local control. To prevent this from occurring, the Local Lockout ('LOCAL LOCKOUT') command disables all front panel controls, including the LOCAL key. The Go to Local ('LOCAL') command will return the instruments to front panel control thereby removing it from the remote and local lockout modes. Note that the above remote and local commands are different from the general HP-IB local and remote commands ('LOCAL 7' and 'REMOTE 7'). The program lines 210-300 exercise the functions 'LOCAL LOCKOUT' and 'LOCAL':

```
210 LOCAL LOCKOUT 7
220 DISP "Local Lockout"
230 PAUSE
240 LOCAL Processor
250 LOCAL Source
260 LOCAL Normalizer
270 LOCAL Testset
280 DISP "Local"
290 BEEP
300 PAUSE
```

Programming Code Conventions

The HP 8505A functions are programmed using a two-character format:

1. The first letter of the control or switch name (for example, "R" for Range), as marked on the front panel of the 8505A.
2. Plus a number indicating the position of the control beginning with 1 at the left or full counter-clockwise position. Thus the 130 MHz RANGE (the second position from the left) is programmed "R2".

Since the 8505A Processor has duplicate controls for each display, a prefix code is used to indicate what channel is being programmed.

"C1" is used for Channel 1.
"C2" is used for Channel 2.
"C3" is used for Electrical length controls.
"C0" is used for the CRT display bandwidth programming.

The letter "E" is used to indicate the end of programming information and must be used when controls are programmed. Refer to Appendix 1 for a table of all 8505A commands.

PROGRAMMING FREQUENCY

There are three recommended ways to program the frequency at which S-Parameter measurements are made: CW $\pm \Delta F$ mode, MARKER/SWEPT mode, and CW mode. Basically, the CW $\pm \Delta F$ ($\Delta F + 0$) mode is the most useful one. The MARKER/SWEPT mode is the fastest while the CW mode has the greatest resolution and potential accuracy.

NOTE

To prevent errors in the Start/Stop frequencies displayed, the WIDTH switch must be programmed in the Data String prior to the START and STOP FREQUENCY MHz registers.

NOTE

Changing the frequency of "FA" or "FB" when in "W3" (ALT Sweep) or entering a frequency >1300 MHz will cause errors in the frequency readout.

In general, when programming the frequency registers, the decimal point is not accepted. Its position is assumed and changes with the frequency range switch setting:

Range 1 (0.5 - 13 MHz) - dd.dd MHz
Range 2 (0.5 - 130 MHz) - ddd.d MHz
Range 3 (0.5 - 1300 MHz) - dddd MHz

For example, "FA1000" programs 10 MHz on Range 1, while "FA0010" programs 10 MHz on Range 3.

Fast CW Mode

For most automatic applications, the CW ± ΔF mode with ΔF set to zero is recommended. In this mode (and all swept modes) the wait to settle within 0.1% of frequency step is 130ms as compared to 1500ms in CW mode (see Table 1).

```
310 ! To set 1113 MHz in "fast CW"
311 !
320 OUTPUT Source;"R3M3W4FB0E"
330 F=1113
340 IMAGE "FA",K,"E"
350 OUTPUT Source USING 340;F
360 WAIT .13
370 DISP "Freq. Set To 1113 MHz"
380 BEEP
390 PAUSE
```

In the above 'IMAGE' statement, "K" specifies that the corresponding data item is to be output with no leading or trailing blanks. This format is required when programming the 8505A. Typically, a program would be structured so that line 320 is executed only once. To program 10.14 MHz, the code would be changed to "R1M3" in line 320 and "F=1014" in line 330. The 'WAIT' time required in line 360 is explained in Table 1.

CW Mode

This mode is programmed as shown in line 450. The actual frequency is programmed by "FAdddd" where dddd is the coarse frequency.

To achieve greater resolution in CW, you may use "FAdddZFBdddd". "FBdddd" is essentially a vernier in CW where 1000 counts in FB equal 2 counts in the Z position of FA. The following example shows how various frequencies could be programmed using the "dddZFBdddd" format.

```
400 ! Set freqs in "CW mode"
401 !
405 DISP "INPUT ANY FREQUENCY ";
410 INPUT "(.5-1300 MHz)",Freq
415 ! R = Range
420 R=1+(Freq>13)+(Freq>130)
425 ! T = Scale Factor
430 T=10^(3-R)
435 ! F = Freq. for 8505
440 F=Freq*T
442 M=INT(F)
445 L=500*(F-INT(F))
450 OUTPUT Source;"M3W5E"
460 IMAGE "R",K,"FA",K,"FB",K,"E"
470 OUTPUT Source USING 460;R,M,L
480 WAIT 1.5
490 DISP "Frequency of";Freq;
495 DISP " MHz set in CW Mode!!"
500 BEEP
510 PAUSE
```

Marker Swept Modes

Displaying the full sweep while taking data at a single frequency is a powerful technique in automatic testing. It allows you to actually "see" the data being taken. It is accomplished by first programming the instrument to either START/STOP 1, START/STOP 2, or CW ± ΔF and then taking data by moving the frequency marker. Any swept mode is programmed by entering the frequency limits into the FA and FB registers.

```
520 ! Sweep between 50 and 100
521 ! MHz,START/STOP,RANGE 2
522 !
525 IMAGE "R2M3W2FA500FB1000E"
530 OUTPUT Source USING 525
540 DISP "START/STOP:";
545 DISP " 50 TO 100 MHz IS SET!"
550 BEEP
560 PAUSE
565 ! Set Center freq=10 MHz with
570 ! del F=0.5 MHz in RANGE 1
571 !
580 IMAGE "R1M3W4FA1000FB500E"
585 OUTPUT Source USING 580
590 DISP "Sweep: CW/del F Mode!"
600 BEEP
610 PAUSE
615 ! Set a 180-210 MHz display
620 ! with the marker at 198 MHz
621 !
625 IMAGE "R3M3W1FA180FB210E"
630 OUTPUT Source USING 625
635 !Mkr posn = % of horiz axis
640 F=(198-180)*100/(210-180)
650 IMAGE "FC",K,"E"
660 OUTPUT Source USING 650;F
670 WAIT .13
675 DISP "START/STOP 1 & MARKER";
680 DISP "Frequencies set!"
690 BEEP
700 PAUSE
710 DISP
```

Once a sweep has been programmed, the frequency marker is positioned on the display using "FCdd", where dd corresponds to screen position in percent of screen (1% being the far left, 99% the far right). The data read will correspond to the value at the frequency marker.

Table 1. Characteristics of Frequency Setting Modes

	CW	Fast CW	Marker Swept
Resolution			1-99%
13 MHz	0.02 kHz	0.01 MHz	
130 MHz	0.2 kHz	0.1 MHz	
1300 MHz	2.0 kHz	1.0 MHz	
Wait time*	1.5 s	130 ms	130 ms
Accuracy	**	1% of range	1% of range

* Time to settle to 0.1% of frequency stepped.
**Can be corrected to counter accuracy ($1/10^5$).

EXAMPLE MEASUREMENT PROGRAM

In the following example, it is assumed that the instrument "addresses" are defined, and that the source has been programmed to the desired START/STOP 1, START/STOP 2, or CW ± ΔF settings. The SCAN TIME should be set to the 100ms position with the VERNIER turned fully clockwise.

The example program will read frequency on Frequency Counter, return loss on CHAN 1 marker, and phase on CHAN 2 marker at 49 points across the CRT. The data measured will then be used to calculate Rho, SWR, and complex load impedance. The program will print out the results and draw plots of Return loss versus frequency and SWR versus frequency. The system printing device is specified by "PRINTER IS 1" (the CRT) or "PRINTER IS 701" (HP 2673A). The "DUMP GRAPHICS" statement copies the contents of the CRT graphics display to a printer specified.

```
720 ! Start "MEASUREMENT" Program
721 !
730 OPTION BASE 1
740 DIM Fr(50),Returnloss(50)
745 DIM Phase(50),Rho(50),Swr(50)
750 DIM Realpart(50),Imagpart(50)
760 IMAGE 1X,5A,6X,6A,5X,5A,7X,3A,9X,3A
770 PRINT USING 760;"FREQ","RETLS.","PHASE",
    "RHO","SWR"
780 PRINT USING 760;"(MHz)","(dB)","(deg)"
```

Set up a 180 to 210 MHz display with the frequency marker at 180 MHz.

```
790 OUTPUT Source;"M3W1T1S4E"
800 OUTPUT Source;"R3FA180FB210E"
810 IMAGE "FC",K,"E"
820 OUTPUT Source USING 810;1
830 WAIT .3
835 IMAGE "C1M2I4D2C2M3I4D2E"
840 OUTPUT Processor USING 835
```

The device under test (DUT) for this example was a 200 MHz band-pass filter. You may change the frequency range according to your specific DUT. The SCAN time "S4" was chosen to give the desired Frequency Counter resolution. In this example, line 840 sets up the 8505A so that CHAN 1 is used for return loss measurements, and CHAN 2 is used for phase measurements.

The following program lines will step the frequency marker through 49 points across the CRT display.

```
850 !Read Frequency,CHAN 1,CHAN 2
855 !marker at 49 points across
860 !the CRT display.
861 !
870 FOR J=1 TO 96 STEP 2
880 IMAGE "FC",K,"E"
890 OUTPUT Source USING 880;J
900 WAIT .3
```

Channel 1 and 2 marker values are automatically measured on alternate sweeps. Because no delays are built into the 8505A, a WAIT statement for two sweeps plus a retrace should be programmed prior to requesting processor readings. With both channels turned on, the WAIT time in milliseconds equals three times the SCANTIME.

```
910 N=1+(J DIV 2)
920 ENTER Source;Fr(N)
930 ENTER Processor;Returnloss(N),Phase(N)
```

The following calculations will reformat the measured data to give Rho, SWR, and complex load impedance.

```
940 !  Calculate RHO
941 !
950 Rho(N)=10^(Returnloss(N)/(20))
951 !
952 !
960 !  Calculate SWR
961 !
970 IF Rho(N)=1 THEN
980 Swr(N)=1000
990 ELSE
1000 Swr(N)=(1+Rho(N))/(1-Rho(N))
1010 END IF
1011 !
1012 !
1020 !  Calculate Load-Impedance
1021 !
1030 Zo=50 !Characteristic Z
1040 DEG
1050 Denom=1-2*Rho(N)*COS(Phase(N))+Rho(N)^2
1060 Realpart(N)=Zo*(1-Rho(N)^2)/Denom
1070 Imagpart(N)=Zo*(2*Rho(N)*SIN(Phase(N)))
     /Denom
```

Print out the results to a specified printer:

```
1080 ! Print formatted output
1081 !
1090 IMAGE 3D.2D,4X,S2D.3D,4X,S3D.2D,4X,2D.3D,
     4X,4D.3D
1100 PRINT USING 1090;Fr(N)/1.E+6,
     Returnloss(N),Phase(N),Rho(N),Swr(N)
1110 NEXT J
1120 BEEP
1130 PAUSE
1140 PRINT USING "2/"
1141 !
1142 !
1150 !  Print Load-Impedance
1151 !
1160 PRINT CHR$(10);"LOAD-IMPED:"
1170 PRINT "(Zo=50 Ohm)";"/"
1180 IMAGE 1X,9A,8X,9A
1190 PRINT USING 1180;"REAL-PART","IMAG-PART"
1200 FOR I=1 TO 48
1210 IMAGE S4D.4D,7X,S4D.4D
1220 PRINT USING 1210;Realpart(I),Imagpart(I)
1230 NEXT I
1240 PRINT USING "3/"
1250 BEEP
1260 PAUSE
```

Initialize display for graphics:

```
1270 !Plot data
1271 !
1280 ALPHA OFF
1290 GCLEAR
1300 GRAPHICS ON
1310 GINIT
```

The following program draws a plot of return loss versus frequency. You may have to change the values in "WINDOW" and "AXES" statements according to the frequency range chosen and return loss measurements. The "DUMP GRAPHICS #701" statement may also have to be modified depending on your printer.

```
1320 ! Plot Returnloss vs. Freq.
1321 !
1330 WINDOW 170,220,-40,5
1340 AXES 5,5,175,0,1,1
1350 ! Label axes
1360 DEG
1370 LDIR 0
1380 MOVE 200,2
1390 LORG 5
1400 CSIZE 4
1410 LABEL "FREQUENCY (MHz)"
1420 LDIR 90
1430 MOVE 172,-20
1440 LABEL "RETURN LOSS (dB)"
1450 LORG 4
1460 FOR I=1 TO 48 STEP 47
1470 MOVE Fr(I)/1.E+6,3
1480 LABEL INT(Fr(I)/1.E+6)
1490 NEXT I
1500 LDIR 0
1510 LORG 5
1520 FOR I=5 TO 30 STEP 25
1530 MOVE 173,-I
1540 LABEL I
1550 NEXT I
1560 ! Plot data
1570 LINE TYPE 4
1580 MOVE Fr(1)/1.E+6,Returnloss(1)
1590 FOR I=1 TO 48
1600 DRAW Fr(I)/1.E+6,Returnloss(I)
1610 NEXT I
1620 DUMP GRAPHICS #701
1630 LINE TYPE 1
1640 OUTPUT 701 USING "6/"
1650 BEEP
1660 PAUSE
```

The following programs draws a plot of SWR versus frequency. Again, you may have to change the values in "WINDOW" and "AXES" statements according to the frequency range chosen and SWR Measurements. In addition, the statement "DUMP GRAPHICS #701" may have to be changed according to the printer in use.

```
1670 ! Plot SWR vs. Frequency
1671 !
1680 GCLEAR
1690 WINDOW 170,220,-10,25
1700 AXES 5,5,175,0,1,1
1710 ! Label axes
1720 MOVE 200,-5
1730 LABEL "FREQUENCY(MHz)"
1740 LDIR 90
1750 MOVE 172,15
1760 LABEL "SWR"
1770 FOR I=1 TO 48 STEP 47
1780 MOVE Fr(I)/1.E+6,-4
1790 LABEL INT(Fr(I)/1.E+6)
1800 NEXT I
1810 LDIR 0
1820 LORG 5
1830 FOR I=5 TO 20 STEP 15
1840 MOVE 173,I
1850 LABEL I
1860 NEXT I
1870 ! Plot data
1880 LINE TYPE 4
1890 MOVE Fr(1)/1.E+6,Swr(1)
1900 FOR I=1 TO 48
1910 DRAW Fr(I)/1.E+6,Swr(I)
1920 NEXT I
1930 LINE TYPE 1
1940 LDIR 0
1950 DUMP GRAPHICS #701
1960 BEEP
1970 PAUSE
1980 GCLEAR
```

PROGRAMMING THE 8503A

Use the following program format for addressing the test set and selecting the S-Paramter Select Switch to either FORWARD - "11" or REVERSE - "2".

Testset = 720
Output Testset; "T1" ! Forward mode
Output Testset; "T2" ! Reverse mode

PROGRAMMING THE 8501A

Graphics

Vector diagrams, lines of text, and rectilinear and polar plots can be displayed by transferring a sequence of display instructions and data from the controller to 8501A memory. The 8501A refreshes the CRT display every 10-20 milliseconds to present a flicker-free display. Memory is available for calculator-generated graphics without sacrificing any of the standard capabilities. Each memory word contains either a display instruction or data. The display instruction identifies how the 8501A processor is to interpret the data which follows. The data consists of a coordinate value or an ASCII character. Refer to Appendix 2 for a table and summary of the HP-IB commands and mnemonics accepted by 8501A.

As shown in Figure 2, up to 22 lines of text, each line up to 54 characters long, can be displayed.

Rectilinear graphs, scaled from the 8505A Channel 1 or Channel 2 reference line, can be plotted on the CRT by writing the "GR" display instruction followed by up to 500 values for y. When the 8501A processor encounters the "GR" instruction in memory, it draws a blank vector (beam off) to the following y value with x=0, then plots the remaining y values automatically incrementing the x value to provide the proper display.

The position of the y=0 point depends upon the Channel 1 reference line as set by the 8505A controls. The y values are integer values ranging from −512 to +512 but if the reference line is at center screen only values between −250 and +250 will be displayed (see Figure 3).



Figure 2. Text Mode Character Positions



Figure 3. GR Cartesian Graph Scaling

The following example shows how the text mode and rectilinear plot can be utilized.

```
1985 ! Display label;
1990 ! Draw a sine wave--8501A
1991 !
2000 ALPHA ON
2010 PRINT USING "5/"
2020 DISP "Label on 8505 CRT"
2030 OUTPUT Normalizer;"ERP6PA112,
     176LBRECTILINEAR PLOT--SINE WAVE"
2040 IMAGE W
2050 OUTPUT Normalizer USING 2040;3
2060 BEEP
2070 PAUSE
2080 DEG
2090 DISP "Sine wave on 8505 CRT"
2095 !Graphics mode
2100 OUTPUT Normalizer;"ERP6GR"
2110 FOR X=1 TO 500
2120 Y=250*SIN(360*X/500)
2130 OUTPUT Normalizer;INT(Y)
2140 NEXT X
2150 BEEP
2160 PAUSE
```

High Speed Binary Transfer

The 9826A Desktop Computer can perform high speed block data transfer both to and from the 8501A memory. Using this capability, both traces (500pts./trace) can be transferred to a disk file in 3 seconds.

```
2170 ! High speed binary transfer
2171 !
2180 PRINT USING "5/"
2190 OUTPUT Source;"M3W1T1S4E"
2200 OUTPUT Source;"R3FA180FB210E"
2210 OUTPUT Normalizer;"ERP6"
2220 LOCAL Normalizer
2230 INTEGER A(1536)
2240 INPUT "FILE NAME?",File$
2250 CREATE BDAT File$,12
2260 OUTPUT Normalizer;"P3,002"
2270 IMAGE %,W
2280 ENTER Normalizer USING 2270;A(*)
2290 ASSIGN @D TO File$
2300 OUTPUT @D;A(*)
2310 ASSIGN @D TO *
2320 PRINTER IS 1
2330 DISP
2340 PRINT "High speed block data transfer
     has been performed!"
2350 PRINT CHR$(10);"Press CONTINUE to
     retrieve the stored data!"
2360 PRINT USING "5/"
2370 BEEP
2380 PAUSE
```

Array "A" is dimensioned to 1536 elements to read three 512-word memory pages. Page 3 is addressed and the "02" command selects the 8501A binary output mode in which data is transferred from memory in a two-byte binary form with the most significant byte first. Then the array "A" is stored under the file name given by the user.

To retrieve data from disc and transfer it to the 8501A memory, use the following sequence.

```
2390 ASSIGN @D TO File$
2400 ENTER @D;A(*)
2410 IMAGE W
2420 OUTPUT Normalizer;"ERP3IB"
2430 OUTPUT Normalizer USING 2410;A(*)
2440 OUTPUT Normalizer USING "B";128
2450 BEEP
2460 PAUSE
```

The file named by "File$" is loaded into memory, then 8501A is initialized using "ER", and Page 3 is addressed to receive data. The "IB" command selects the 8501A binary input mode in which data is received in a two-byte binary format, most significant byte first. The statement 2440 sends 128 in a binary form, which returns the 8501A to ASCII mode.

PHASE LOCK OPTION SOFTWARE CONSIDERATION

The applications of phase lock can utilize two different methods of operating the stable source (synthesizer or generator):

SWEPT OPERATION

The stable source is tuned to the center frequency of the device under test and deviated (up to 130 kHz) via the ΔF control on the 8505A. Parameter measurements (magnitude, phase, or group delay) are obtained by moving the FC marker across the screen.

CW OPERATION

The ΔF is set to zero on the 8505A and the center frequency of the synthesizer is stepped. The center frequency of the 8505A Source is set initially, and as long as the synthesizer is within the capture range of the phase lock loop, the 8505A Source need not be retuned.

One can determine whether phase lock has been achieved by observing or reading the 8505A Frequency Counter. It should be identical to the synthesizer frequency setting.

For general description and operation instructions on phase lock, refer to 8505A - Option Supplement Chapter E (HP Supplement Part No. 08505-90070). (See Figure 4 for system connections.)

The following program assumes the 8505A Source has address 719 and the 8660C Synthesizer address 718. Set up the Source for Range 3, LIN EXPAND, CW Mode with +10 dBm, and Phase Lock.

```
2465 ! Synthesizer programming
2470 ! routine for
2480 ! HP 8660C.
2481 !
2490 OUTPUT Source;"R3M3W4E"
2495 ! +10dBm P.Lock
2500 OUTPUT Source;"08V99M4E"
```

Set up a sweep range.

```
2510 INPUT "START FREQ (MHz)",F1
2520 INPUT "STOP FREQ (MHz)",F2
2530 INPUT "STEP SIZE (MHz)",F3
2540 IMAGE "FA",K,"FB",K,"E"
2550 OUTPUT Source USING 2540;F1,0
2560 I=((F2-F1)/F3)+1
2570 F5=F1
```

Set the synthesizer center frequency to 200 MHz and the output level to +3 dBm.

```
2580 OUTPUT Synthesizer;"/20(-100C"
2590 DISP "CF=200 MHz AT +3dBm"
2600 WAIT 3
```

In order to satisfy the internal logic requirements of the 8660C, the least significant digits and the most significant digits of frequency setting must be exchanged. The function "FNInv" will do this arrangement quickly, efficiently, and automatically.

```
2610 FOR N=1 TO I
2620 F=F5
2630 IMAGE "FA",K,"E"
2640 OUTPUT Source USING 2630;F
2650 F4=FNInv(F)
2660 IMAGE "/",K,"("
2670 OUTPUT Synthesizer USING 2660;F4
2680 WAIT .5
2690 DISP "PLEASE PRESS CONTINUE"
2700 PAUSE
2710 DISP
2720 F5=F5+F3
2730 NEXT N
2740 DISP "FREQ RANGE SWEPT!"
2750 END
2751 !
2760 !Inversion routine!
2761 !
2770 DEF FNInv(P1)
2780 DIM A$[10],B$[10]
2790 OUTPUT A$ USING "#,10Z";P1*1.E+6
2800 FOR I=1 TO 10
2810 B$[I,I]=A$[11-I,11-I]
2820 NEXT I
2830 RETURN VAL(B$)
2840 FNEND
```

Data can now be taken as before with the frequency marker or via the high speed binary transfer routine with the 8501A.



Figure 4. Equipment Set-up for Phase-Lock Operation

APPENDIX 1

8505A Signal Processor

Function and Comment	HP-IB Code
IF and VIDEO filter selection*	C0Bd
CHANNEL code must precede the other codes used in the Channel. C1 codes are the same as C2 codes.	C2
INPUT*	Id
MODE*	Md
SCALE/Div*	Sd
REFerence OFFSET	R±ddddd
STORED CALIBRATION	O±ddddd
The range of numbers for "R" and "O" is ±19999. The plus sign is implied and leading zeros ignored. Decimal position is not programmed. It is shown below to give implied position.	
ddddd Phase (degrees) Delay 1300 MHz (ns) Magnitude (dB) Delay 13 MHz (μs) Delay 130 MHz (μs)	
DISPLAY* D3 when programmed clears only "R", to clear "O" enter O. Non-Learned programming code.	Dd

Function and Comment	HP-IB Code
ELECTRICAL LENGTH code must precede the other codes in C3.	C3
INPUT*	ID
MODE*	Md
ELECTRICAL LENGTH	R±ddd
Stored Electrical Length	O±ddd
The range of numbers for "R" and "O" is ±199. The plus sign is implied and leading zeros ignored. Decimal position is not programmed. It is shown below to give implied position.	

ddd	Source Range	C3 Mode	Metric Scale
	1	3	m
	3	3	cm
	1	2	m
	2	3	m
	3	2	cm
	2	2	m

In C3M4 Mode the scale is in tens of degrees per scan with a range of ±1700 deg.	
DISPLAY* D2 when programmed does not clear stored Length to clear this enter O0. Non-Learned programming code.	Dd
Mode of unselected input only on Learn string (characters 44 and 45). Non-Programmable.	Sd
Terminator used to end and separate all ("R", "O", and "D") program statements.	E

8505A Frequency Source

Function and Comment	HP-IB Code
START FREQUENCY and CW	FAdddd
STOP FREQUENCY and ΔF	FBdddd
Leading zeros are ignored. Decimal position is not programmed. It is shown below to give implied position. The range of frequencies for Range and Mode settings are shown below.	

ddddRange	Mode 1	Mode 2 & 3
3	1-1000 MHz	1-1300 MHz
2	1-100.0 MHz	.5-130.0 MHz
1	1-10.00 MHz	.5-13.00 MHz

Vernier controls default to min (CCW) when Local Lockout is set. In an Option 005 instrument using MODE "M4", the units for FB change to KHz.	
FREQUENCY COUNTER Only one marker available in remote. Set marker (dd) to percentage of sweep width between 00% and 99%.	FCdd

Function and Comment	HP-IB Code
OUTPUT LEVEL dBm 10 dB steps*	Od
VERNIER Local Lockout required to program. Setting not on Learn String.	
dd = 99 ((V + 12)/12)	Vdd
V = Vernier in dB	Id
INPUT LEVEL dBm MAX*	
RANGE MHz*	Rd
MODE (sweep)* M4 = PHASE LOCK mode for Option 005 instruments	Md
WIDTH*	Wd
SCAN TIME SEC* VERNIER defaults to min Scan Time when Local Lockout is set	Sd
TRIGGER*	Td

APPENDIX 2

8501A Mnemonics (1 of 2)

Memory Mnemonics	
MODE SELECTION*	
GR	Graph mode. 500 y values, $-500 \leqslant y \leqslant 500$. $y = 0$ at Channel 1 reference line.
PA	Vector mode. X, y pairs, $0 \leqslant x \leqslant 432$, $0 \leqslant y \leqslant 360$. 0, 0 at lower left-hand corner.
PR	Polar mode. 250 x, y pairs, $-500 \leqslant x, y \leqslant 500$. 0, 0 at polar beam center.
LB	Label mode. Display ASCII and special characters. 22 lines, 50 characters/line. Clear text mode by End-of-Text character. (ASCII 3 character code).
BEAM CONTROL	
PU	Beam off. Blank vector to next point.
PD	Beam on.
MEMORY CONTROL*	
BO	Block off. No data display until next memory mnemonic.
EF	End-of-file. Skip to word 1 of page 1.
PS	Page skip. Skip to word 1 of next page.
ALTERNATE GRAPHICS DISPLAY (PLACE PRECEDING MODE SELECTION MNEMONIC)	
D1	Select PA reference and scaling.
D2	Select GR reference and scaling. $y = 0$ at Channel 1 reference line.
D3	Select GR reference and scaling. $y = 0$ at Channel 2 reference line.
D4	Select PR reference and scaling.

* Mode selection and memory control mnemonics have the effect of AB processor command when input to 8501A.

APPENDIX 2
8501A Mnemonics (2 of 2)

Input/Output

Px, n	Address memory to specify source or destination of data and select ASCII output format. x = page = 1 – 8, n = word = 0 – 511
OL	Output Learn String, 24 characters
OA	Output current memory address. Data output as: page no. comma word no. carriage return-line feed
IB	Input binary format data. (2 byte, high byte first). Place following memory address. Reset to default ASCII mode using HP-IB Clear command or sending most significant bit of high byte line.
01	Output ASCII format data (default). Each data value is: sign integer carriage return-line feed
02	Output binary format data 2 byte, high byte first Place following Memory address.

Front Panel

S1	STORAGE OFF
S2	STORAGE ON
S3	STORAGE HOLD
L1	LABELS OFF
L2	LABELS ON
M1	MAGNIFIER X1
M2	MAGNIFIER X2
M3	MAGNIFIER X5
M4	MAGNIFIER X10
ER	ERASE
C1	CHANNEL 1
C2	CHANNEL 2
I1	INPUT OFF
I2	INPUT ON
I3	INPUT—MEM
MS	MEMORY STORE
V0	MEMORY VIEW OFF
V1	MEMORY VIEW ON
A1	AVERAGING OFF
A2	AVERAGING ON
F1	AVERAGING FACTOR 2
F2	AVERAGING FACTOR 4
F3	AVERAGING FACTOR 8
F4	AVERAGING FACTOR 16
F5	AVERAGING FACTOR 32
	AVERAGING FACTOR 64
	AVERAGING FACTOR 128
	AVERAGING FACTOR 256

Processor Commands

AB	Stop processing data from 8505 A.
EX	Restart processing data from 8505 A.
TSn,	Take n 8505 A sweeps, 1 sweep/channel.

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Southern** (404) 955-1500; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe**: Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan**: Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

5952-9329

Printed in U.S.A.

DESIGNED FOR HP-IB SYSTEMS

Programming Note

8505A/0000-99A January 1980

Supersedes: 8505A/0000-99

Programming Code Summary for 8505A RF Network Analyzer

INTRODUCTION

This programming note contains a summary of the HP-IB codes used in the 8505A and their functions. It is intended for use by those familiar with both the 8505A and HP-IB programming. Refer to the 8505A Operating and Service manual for a complete explanation of the codes and functions.

ADDRESS TABLE

Instrument	Talk	Listen	Decimal
8505A Processor	P	0	16
8505A Source	S	3	19

The addresses may be changed by removing instrument covers and setting appropriate slide switches on the HP-IB circuit boards. For detailed instructions see Chapter A, Section II of the 8505A Operating and Service Manual, HP part No. 08505-90072.

DATA

The 8505A consists of two separately programmed instruments, Processor and Source.

8505A Instrument Conventions

1. Program letter codes must be uppercase, leading zeros and spaces are ignored.
2. Any controls not programmed will assume their front panel state as positioned before remote.
3. The switches are programmed using two character format.
   a. The first character is a letter corresponding to switch name (example "R" for range).
   b. The second character is a number corresponding to the position of the switch beginning with 1 at the left or CCW position.
   c. Other controls have the same alpha-numeric sequence but may use a two letter code, a plus or minus sign, and up to a 5 digit number code.

Processor Programming Conventions

1. The Processor codes may be sent in any order except:
   a. The duplicate controls for each display channel require that the following prefix codes be used.
      "C1" for all CHANNEL 1 codes
      "C2" for all CHANNEL 2 codes
      "C3" FOR ALL ELECTRICAL LENGTH codes
      "C0" for the BandWidth code
   b. Use the letter "E" to separate all "R", "O", and "D" program statements and to end the programming string.
2. The last digit programmed in REFerance OFFSET may not be displayed on the front panel LEDs.

Source Program Convention

1. The Source codes may be sent in any order except:
   a. Range "R", Mode "M", and Width "W" codes should precede "FA" and "FB" codes.
   b. The letter "E" is used to end the programming string.
2. In Option 005 instruments when WIDTH is in CW ± ΔF (W4) and when Mode is in Phase Lock (M4), the ΔF (FB) range and delay resolution are reduced by a factor of 1000.

Reading the 8505A

The Processor MARKER values may both be read by reading them into two variables. The units are the same as displayed on the front panel LEDs.

The Counter marker value is obtained by reading the source into a variable. The units are in Hz.

The current state of many of the 8505A controls may be output using the Learn Mode (program "L") then read on a properly dimensioned string.

Processor Learn String: 85 Characters

```
"COBd, C1Id Md Sd R ± ddddd, C2Id Md Sd
       4    9  11 13  15          25 27 29
R ± ddddd, C3Id Md Sd R ± ddd, E,
 31          41 43 45  47
C1O ± ddddd, C2O ± ddddd, C3O ± ddd, E"
     57           67          77    82
```

Source Learn String: 30 characters

```
"Od Id Rd Md Wd Sd Td,FAdddd,FBdddd,E"
  2  4  6  8  10 12 14     20    25   30
```


HEWLETT PACKARD





A:C0B1
B:C0B2
A+C:C0B3
B+C:C0B4

NOT PROGRAMMABLE



PROCESSOR PROGRAMMING CODES

FUNCTION AND COMMENT	HP-IB CODE
IF and VIDEO filter selection*	C0Bd
CHANNEL code must precede the other codes used in the channel. C1 codes are the same as C2 codes. INPUT* MODE* SCALE/DIV* REFerence OFFSET STORED CALIBRATION The range of numbers for "R" and "O" is ± 19999. The plus sign is implied and leading zeros ignored. Decimal position is not programmed. It is shown below to give implied position. d d d d d Phase (degrees) Delay 1300 MHz (ns) Magnitude (dB) Delay 13 MHz (μs) Delay 130 MHz (μs)	C2 Id Md Sd R ± ddddd O ± ddddd
DISPLAY* D3 when programmed clears only "R", to clear "O" enter O. Non-learned programming code.	Dd
*The range for values of "d" are shown on corresponding control in illustration above.	

FUNCTION AND COMMENT	HP-IB CODE
ELECTRICAL LENGTH code must precede the other codes in C3. INPUT* MODE* ELECTRICAL LENGTH Stored Electrical Length The range of numbers for "R" and "O" is ± 199. The plus sign is implied and leading zeros ignored. Decimal position is not programmed. It is shown below to give implied position. d d d Source Range / C3 Mode / Metric Scale: 1 — 3 — m 3 — 3 — cm 1 — 2 — m 2 — 3 — m 3 — 2 — cm 2 — 2 — m In C3M4 Mode the scale is in tens of degrees per scan with a range of ± 1700 deg. DISPLAY* D2 when programmed does not clear stored Length to clear this enter O0. Non-Learned programming code.	C3 Id Md R ± ddd O ± ddd Dd
Mode of unselected input only on Learn string (characters 44 and 45). Non-Programmable.	Sd
Terminator used to end and separate all ("R","O", and "D") program statements	E







SOURCE PROGRAMMING CODES

FUNCTION AND COMMENT	HP-IB CODE
OUTPUT LEVEL dBm 10 dB steps* VERNIER Local Lockout required to program. Setting not on Learn String. dd = 99 ((V + 12)/12) V = Vernier in dB INPUT LEVEL dBm MAX*	Od Vdd Id
RANGE MHz* MODE (sweep)* M4 = PHASE LOCK mode for Option 005 instruments WIDTH* SCAN TIME SEC* VERNIER defaults to min Scan Time when Local Lockout is set TRIGGER*	Rd Md Wd Sd Td
*The range of values for "d" are shown on corresponding control in illustration above.	

FUNCTION AND COMMENT	HP-IB CODE
START FREQUENCY and CW STOP FREQUENCY and ΔF Leading zeros are ignored. Decimal position is not programmed. It is shown below to give implied position. The range of frequencies for Range and Mode settings are shown below. dddd Range Mode 1 Mode 2 & 3 3 1 – 1000 MHz 1 – 1300 MHz 2 1 – 100.0 MHz .5 – 130.0 MHz 1 1 – 10.00 MHz .5 – 13.00 MHz Vernier controls default to min (CCW) when Local Lockout is set. In an Option 005 instrument using MODE "M4", the units for FB change to KHz.	FAdddd FBdddd
FREQUENCY COUNTER Only one marker available in remote. Set marker (dd) to percentage of sweep width between 00% and 99%.	FCdd

SETTLING TIME

OPERATION		DELAY REQUIRED
PROCESSOR settling times		
Processor Reading	1 channel	2 sweep times
C0B BandWidth	2 channels	3 sweep times
10kHz		10 ms
1 kHz		25 ms
Video Filter		300 ms
C1-C2 CHANNELs 1 & 2		
Any change affecting Reference Level (includes REF OFFSET, CLR, INPUT, MODE, and 8503A/B Test Set switching		
To 1% final value		1000 ms
To .01% final value		3000 ms
ZRO (MKR & REF mode), CLR, INPUT, MODE, and SCALE/DIV		At least 3 sweep times per channel + 3000 ms
C3 ELECTRICAL LENGTH		
Any change affecting LENGTH (includes LENGTH, CLR, Channel 1 and 2 MODE, INPUT, and 8503A/B Test Set switching).		3000 ms
ZRO, CLR, INPUT, AND MODE		20 ms
Read marker value and determine if auto-ranging occurs (resolution changes):		
Resolution constant		1 sweep/Channel + 25 ms
Autoranging		2 sweeps/Channel + 100 ms
SOURCE settling times		
Counter Reading		2 sweep times
MAX SCAN-TIME SEC, TRIGGER, OUTPUT, INPUT LEVEL dBm		20 ms
START/STOP, $\pm \Delta F$		120 ms
CW (to 0.01%)		1000 ms (first freq.) 360 ms (next freq.)

OTHER HP-IB COMMANDS

Trigger: The 8505A does not respond to a device Trigger.

Clear: The 8505A does not respond to a device Clear.

Remote: The Remote message will disable the 8505A front panel programmable controls. It will retain the pre-remote front panel settings until changed by program.

Local: The local message or switching the 8505A TRIGGER switch to LOCAL (only if local lockout is not enabled) will set the 8505A to manual front panel control.

Local Lockout: Disables local switch (TRIGGER) on 8505A front panel, "presets" Vernier controls as shown on following table, and enables FB to be used as a Vernier when 8505A is in CW Mode.

VERNIER CONTROL	"PRESET" POSITION
Output Level	– 10 dBm
Scan time	Vernier Scan time min
Frequency (2)	0 Position MAX CCW
Electrical Length	0 Position MAX CCW

Serial Poll Enable: HP-IB, Octal 030.

Serial Poll Disable: HP-IB, Octal 031

Service Request: If the Serial Poll is enabled the Source will Service Request if loss of phase lock or RF input overload occurs. Bit 6 of the Status Byte will also refresh this. An initial Serial Poll after power on will enable the Service Request.

Pass Control: The 8505A does not have the ability to Pass Control or take Control.

Abort: The 8505A does not respond to the Abort message.

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Southern** (404) 955-1500; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write, Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

5952-9293

PRINTED IN USA

DIGITALY REMASTERED
OUT OF PRINT
TEST EQUIPMENT MANUAL SCANS

By

ArtekManuals

(formerly known as ArtekMedia and Artekmedia.com)
7102 VALRIE LANE, RIVERVIEW, FL 33569
www.artekmanuals.com
"High resolution scans of obsolete technical manuals"

If your looking for a quality scanned technical manual in PDF format please visit our WEB site at www.ArtekManuals.com or drop us an email at Manuals@ArtekManuals.com

If you don't see the manual you need on the list drop us a line anyway we may still be able to point you to other sources. If you have an existing manual you would like scanned please write for details. This can often be done very reasonably in consideration for adding your manual to our library.

Typically the scans in our manuals are done as follows;

1) Typed text pages are typically scanned in black and white at 300 dpi.
2) Photo pages are typically scanned in gray scale mode at 450 dpi
3) Schematic diagram pages are typically scanned in black and white at 600 dpi or Gray-scale @450dpi depending on the characteristics.
4) All manuals are text searchable and all manuals are fully bookmarked

All data is guaranteed for life (yours or mine ... which ever is shorter). If for ANY REASON your file becomes corrupted, deleted or lost, ArtekManuals will replace the file for the price of shipping, or free via FTP download.

Thanks

Founder and CEO
Artek Media
ArtekManuals
Outsource-Options